



INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO



Atribuição-NãoComercial-Compartilhalgual 3.0 Brasil (CC BY-NC-SA 3.0)

Você tem a liberdade de:



Compartilhar — copiar, distribuir e transmitir a obra.

Remixar — criar obras derivadas.

Sob as seguintes condições:



Atribuição — Você deve creditar a obra da forma especificada pelo autor ou licenciante (mas não de maneira que sugira que estes concedem qualquer aval a você ou ao seu uso da obra).



Uso não comercial — Você não pode usar esta obra para fins comerciais.



Compartilhamento pela mesma licença — Se você alterar, transformar ou criar em cima desta obra, você poderá distribuir a obra resultante apenas sob a mesma licença, ou sob uma licença similar à presente.

Ficando claro que:

Renúncia — Qualquer das condições acima pode ser **renunciada** se você obtiver permissão do titular dos direitos autorais.

Domínio Público — Onde a obra ou qualquer de seus elementos estiver em **domínio público** sob o direito aplicável, esta condição não é, de maneira alguma, afetada pela licença.

Outros Direitos — Os seguintes direitos não são, de maneira alguma, afetados pela licença:

- Limitações e exceções aos direitos autorais ou quaisquer **usos livres** aplicáveis;
- Os **direitos morais** do autor;
- Direitos que outras pessoas podem ter sobre a obra ou sobre a utilização da obra, tais como **direitos de imagem** ou privacidade.

Aviso — Para qualquer reutilização ou distribuição, você deve deixar claro a terceiros os termos da licença a que se encontra submetida esta obra. A melhor maneira de fazer isso é com um link para esta página.

Sumário

1. Introdução.....	5
1.1. Lista de exercícios 01.....	6
1.2. Lista de exercícios 02.....	7
1.3. Lista de exercícios 03.....	9
1.4. Lista de exercícios 04.....	11
2. Algoritmos e fluxogramas.....	14
2.1. Lista de exercícios 01.....	15
2.2. Lista de exercícios 02.....	21
2.3. Lista de exercícios 03.....	23
2.4. Lista de exercícios 04.....	25
3. Estrutura de repetição, vetores e matrizes.....	29
3.1. Lista de exercícios 01.....	30
3.2. Lista de exercícios 02.....	34
3.3. Lista de exercícios 03.....	38
3.4. Lista de exercícios 04.....	41
3.5. Lista de exercícios 05.....	45
3.6. Lista de exercícios 06.....	48



Fundamentos e Infraestrutura de Redes de Computadores

FIR-Lab-01 → Configurar o endereço IP e a máscara de rede de cada host para que o mesmo possa comunicar-se com outros hosts da rede.

FIR-Lab-02 → Configurar o endereço IP e a máscara de rede de cada host de modo a criar duas redes separadas, uma para a Contabilidade e outra para o Recursos Humanos.

FIR-Lab-03 → Configurar o serviço de DHCP (Dynamic Host Configuration Protocol) no roteador de modo que todas os hosts possam obter as configurações de IP e máscara automaticamente.

FIR-Lab-04 → Configurar o serviço de NAT (Network Address Translation) no roteador R1 de modo que os hosts da Rede LAN que usam IPs privados possam se comunicar com os hosts da Rede WAN por meio de IP's públicos

FIR-Lab-05 → Configurar o DNS no computador PC_1 para que o mesmo possa conectar-se a sites na Internet usando a resolução de nomes DNS.

FIR-Lab-06 → Verificar a criação de sockets na camada de transporte

FIR-Lab-07 → Verificar o funcionamento do protocolo ARP (Address Resolution Protocol).

Comunicação de Dados

CDD-Lab-01 → Verificar a aprendizagem do switch.

CDD-Lab-02 → Verificar a aprendizagem do switch.

CDD-Lab-03 → Configurar o switch de modo a criar duas Virtual LAN (VLAN), uma para a Contabilidade denominada VLAN 10 e outra para o Recursos Humanos denominada VLAN 20.

Telecomunicações

TLC-Lab-01 → Configurar rotas estáticas nos roteadores.

TLC-Lab-02 → Configurar rotas dinâmicas nos roteadores.

TLC-Lab-03 → Configurar uma rede Frame Relay.



1. Introdução

Os laboratórios desta seção tem o objetivo de

Introdução à arquitetura de computadores

Sistemas de numeração

Tipos primitivos de dados

Operadores aritméticos, relacionais e lógicos, precedência de operadores e expressões

Identificadores, constantes, variáveis e comandos de atribuição



1.1. Lista de exercícios 01

1. Converta os seguintes números de decimal para binário:
 - a. 12
 - b. 203
 - c. 34
 - d. 123
 - e. 72

2. Converta os seguintes números de binário para decimal:
 - a. 10010011
 - b. 00110011
 - c. 01010101
 - d. 10001110
 - e. 10001001

3. Um computador conectado à Internet possui o seguinte endereço IP escrito na notação decimal:

200.204.0.15
 - a. Reescreva este endereço na notação binária.

4. Um computador conectado à Internet possui o seguinte endereço IP escrito na notação binária:

11010010.101000.1100.10111
 - a. Reescreva este endereço na notação decimal.

5. Sabendo-se que na Tabela ASCII o caracter "A" equivale ao número 65 em decimal, descubra quais palavras estão escritas nas seqüências abaixo:
 - a. 010011000100010101001001
 - b. 01000110010000010100110101000001
 - c. 0101011001000101010100100100010001000101

6. Um músico fez um programa que usa 3 bits para armazenar as 7 notas musicais: DÓ, RÉ, MI, FÁ, SOL, LÁ e SI. Considerando que a primeira nota equivale ao número 1 em decimal, descubra quais notas estão escritas abaixo:
 - a. 001010011100001010001010001101100011001010011100
 - b. 001111011100111011010011011110101100001111010011



1.2. Lista de exercícios 02

1. Calcule as seguintes expressões:

- a. $5/2$
- b. $5 \text{ DIV } 2$
- c. $5 \text{ MOD } 2$
- d. $2*3+7$
- e. $2*(3+7)$
- f. $7^2/3$
- g. $12/3$
- h. $12 \text{ DIV } 3$
- i. $12 \text{ MOD } 3$

2. Reescreva as seguintes equações na forma linear:

a. $3 \times \frac{5}{8} + 7^4$

b. $\left\{ 12 \times \left[3 - \left(\frac{5}{8} \right)^3 \right] \right\}$

c. $1 - \sqrt{\frac{3}{2}} \times (3 - 4)^2$

d. $\frac{2\sqrt{\frac{3}{12^4}} - 4^{5-3}}{2 \times 2^3 - 3}$

e. $\frac{4^3 + \left(\frac{5}{3} - \frac{7}{2^6} \right)^{3-2}}{\frac{3+2}{12}}$

3. Sabendo-se que $x=3$ e $y=2$, avalie as expressões e verifique se são Verdadeiras ou Falsas:

- a. $x==y$
- b. $x!=(y+1)$
- c. $x>3$
- d. $x>=3$
- e. $(2*x-2)==(y^2)$

4. Sabendo-se que $x=7$ e $y=2*3$, avalie as expressões e verifique se são Verdadeiras ou Falsas:

- a. $(x==y) \mid (x>y)$
- b. $(x\%2)==(y\%2)$
- c. $((x-1)==y) \ \& \ (x==2*y)$
- d. $(x!=y) \mid (x>y) \mid ((x-1)==(y+0))$
- e. $(x!=(y+2)) \ \& \ ((x>=(y+1)) \mid ((x+3)==(y+3)))$

INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO



5. Uma máquina de bebidas quentes pode fornecer, a partir de água quente (AQ), café solúvel (CS) e açúcar (AC), as seguintes bebidas: café amargo (CA) e café doce (CD). Quais expressões abaixo são válidas:
- AQ & CS
 - AQ & AC
 - AC & CS
 - AQ & CS & !AC
 - AQ & CS & AC
 - !AQ & CS & AC
 - AQ & !CS & AC
 - AQ | CS | AC
6. Uma máquina de bebidas quentes pode fornecer, a partir de água quente (AQ), café solúvel (CS), chocolate em pó (CP) e açúcar (AC), as seguintes bebidas: café amargo (CA), café doce (CD) e chocolate quente (CQ). Considerando que no chocolate quente não vai açúcar, quais expressões abaixo são válidas:
- AQ & CP
 - AQ & AC & CP
 - AQ & CS & !AC
 - AQ | CS | CP
 - !AQ & CP
 - AQ & AC & (CP | CS)
 - AQ & (CS | CP)
 - AQ & (CS & CP)
7. Uma financeira oferece empréstimos bancários com taxas de juros que variam de acordo com o perfil do cliente. Para aposentados (AP), a taxa de juros é de 1% ao mês; para funcionários públicos (FP) é de 2%; assalariados (AS) 3% e autônomos (AU) 4%. Sabendo-se que a financeira só empresta dinheiro para assalariados com pelo menos 3 meses de carteira assinada e para autônomos desde que possuam talão de cheques, avalie quais expressões abaixo são válidas:
- AP
 - AS & (1 mês)
 - AP | FP
 - FP & cheques
 - AU & !cheques
 - AS & (carteira assinada > 3 meses)
 - AS & (carteira assinada >= 3 meses)
 - FP & !cheques
 - AP | FP | AU
 - AP | FP | (AU & cheques)

Lista de operadores:

Aritméticos: (e); ^ e sqrt(); *, /, DIV e MOD; + e -

Relacionais: >, <, >=, <=, == e !=

Lógicos: &, | e !



1.3. Lista de exercícios 03

1. Fazer um programa que leia dois números inteiros e calcule a divisão entre eles.

<pre>algoritmo "Inteiros" var X, Y, R: inteiro inicio leia(X) leia(Y) R <- X/Y escreva(R) fimalgoritmo</pre>	Discuta os resultados, problemas e aponte possíveis soluções:
---	---

2. Usar a estrutura do programa acima para testar os operadores de Divisão Inteira e Resto de Divisão. Explicar a diferença entre os dois operadores:

3. Refazer o programa do Exercício 1 com números Reais (Fracionários), para testar os operadores de Divisão Inteira e Resto de Divisão. Discuta os resultados, problemas e aponte possíveis soluções.

4. Fazer um programa que calcule as seguintes expressões:

<pre>algoritmo "Expressoes" var R: real inicio R <- <expressão> escreva(R) fimalgoritmo</pre>	<p>1. a) $3 \times \frac{5}{8} + 7$</p> <p>2. b) $\left\{ 12 \times \left[3 - \left(\frac{5}{8} \right) \right] \right\}$</p>	<p>3. c) $1 - \sqrt{\frac{3}{2}} \times (3 - 4)$</p> <p>4. d) $\frac{4 + \left(\frac{5}{3} - \frac{7}{2} \right)}{\frac{3 + 2}{12}}$</p>
--	---	---



5. Fazer um programa que avalie as seguintes expressões:

<pre>algoritmo "Logica" var X, Y, R: logico inicio X <- VERDADEIRO Y <- FALSO R <- <expressão> escreva(R) fimalgoritmo</pre>	<p>5.</p> <p>6. a) $X==Y$</p> <p>7. b) $X!=Y$</p> <p>8. c) $!X$</p> <p>9. d) $!Y$</p>	<p>10.</p> <p>11. e) $(X==Y) (X!=Y)$</p> <p>12. f) $(X==Y) \& (X!=Y)$</p> <p>13. g) $X \& Y$</p> <p>14. h) $X Y$</p>
---	---	--

6. Fazer um programa que avalie as seguintes expressões:

<pre>algoritmo "Relacionais" var X, Y: inteiro R: logico inicio X <- 5 Y <- 7 R <- <expressão> escreva(R) fimalgoritmo</pre>	<p>15.</p> <p>16. a) $X==Y$</p> <p>17. b) $X>Y$</p> <p>18. c) $(X+2)>Y$</p> <p>19. d) $(X+2)>=Y$</p>	<p>20.</p> <p>21. e) $(X+1)==(Y-1)$</p> <p>22. f) $(X<Y) \& (X==Y)$</p> <p>23. g) $(X<Y) \& (Y>X)$</p> <p>24. h) $(2X)==(2Y-4)$</p>
---	---	--

7. EXERCÍCIO EXTRA – O programa abaixo é uma aplicação prática do uso do operador de Resto de Divisão. Explique o que este programa faz:

<pre>algoritmo "ParImpar" var X, R: inteiro inicio escreva("Digite um numero inteiro: ") leia(X) R <- X MOD 2 se R=0 entao escreva("O numero ", X, " é PAR. ") senao escreva("O numero ", X, " é IMPAR. ") fimse fimalgoritmo</pre>	
--	--



1.4. Lista de exercícios 04

1. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Inteiros"  
02 var  
03   X, Y: inteiro  
04   R: real  
05 inicio  
06   leia(X)  
07   leia(Y)  
08   R <- X/Y  
09   escreva(R)  
10 fimalgoritmo
```

2. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Inteiros_2"  
02 var  
03   X, Y: inteiro  
04   R: real  
05 inicio  
06   escreva("Digite um numero: ")  
07   leia(X)  
08   escreva("Digite outro numero: ")  
09   leia(Y)  
10   R <- X/Y  
11   escreva(X, " dividido por ", Y, " eh igual a: ", R)  
12 fimalgoritmo
```

3. Quais são as diferenças entre os programas dos exercícios 1 e 2?

4. Explique detalhadamente o que acontece na linha 11 do programa do exercício 2:

5. No programa do exercício 2, troque os nomes das variáveis X, Y e R por nomes mais convenientes e que estejam de acordo com as regras de nomeação de variáveis.



6. Analisar e testar o código de programa abaixo:

```
01 algoritmo "AreaRetangulo"
02 var
03     LadoA, LadoB, AreaRetangulo: inteiro
04 inicio
05     escreva("Digite o comprimento do Lado A: ")
06     leia(LadoA)
07     escreva("Digite o comprimento do Lado B: ")
08     leia(LadoB)
09     AreaRetangulo <- LadoA*LadoB
10     escreva("A area do retangulo eh: ", AreaRetangulo)
11 fimalgoritmo
```

7. O que acontece se acrescentarmos o comando `escreval` após a linha 09 do programa do exercício anterior?

8. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Media"
02 var
03     Nota1, Nota2, Nota3, Nota4: inteiro
04     Media: real
05 Inicio
06     escreva("Digite a primeira nota (entre 0 e 10): ")
07     leia(Nota1)
08     escreva("Digite a segunda nota (entre 0 e 10): ")
09     leia(Nota2)
10     escreva("Digite a terceira nota (entre 0 e 10): ")
11     leia(Nota3)
12     escreva("Digite a quarta nota (entre 0 e 10): ")
13     leia(Nota4)
14     Media <- (Nota1+Nota2+Nota3+Nota4)/4
15     escreval
16     escreva("A media eh igual a: ", Media)
17 fimalgoritmo
```

9. No final da linha 16 do programa do exercício anterior, o que acontece se trocarmos `Media` por `Media:2:1`? E se trocarmos por `Media:2:6`? E `Media:10:4`?

INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO



10. Para os programas dos exercícios 2, 6 e 8, identificar a entrada, o processamento e a saída.



2. Algoritmos e fluxogramas

Os laboratórios desta seção tem o objetivo de

Algoritmos, fluxogramas e codificação
Estruturas de decisão



2.1. Lista de exercícios 01

PARTE I

Um cliente chega até uma farmácia com uma receita médica à procura de um remédio. Do ponto de vista do atendente, descreva os passos necessários para atender o cliente desde a verificação da receita até o pagamento no caixa.

Algoritmo:


ALTO NÍVEL	BAIXO NÍVEL
1. Receber cliente	1. Recebe o cliente;
2. Verificar receita	2. Verifica a receita médica e identifica as prescrições;
	3. Se a receita for de remédio controlado, pedir um documento de identificação do paciente e reter uma via da receita;
3. Verificar disponibilidade	4. Caso o cliente se negue a apresentar documento de identificação ou a fornecer a segunda via da receita, encerrar o atendimento;
	5. Localiza o(s) remédio(s) na prateleira;
	6. Caso não exista(m) o(s) remédio(s) na prateleira, verificar no estoque;
	7. Caso não exista(m) o(s) remédio(s) no estoque, verificar se o cliente aceita substituí-lo(s) por genérico(s), quando for o caso e se estiver disponível;
4. Informar procedimentos de uso	8. Caso o cliente não aceite, encerrar o atendimento;
	9. Caso existam apenas alguns dos remédios prescritos na receita, verificar se o cliente deseja fazer a comprar parcial, caso contrário, encerrar o atendimento;
5. Receber pagamento	10. Informar o(s) procedimento(s) de uso do(s) remédio(s), a dose diária e a duração do tratamento;
	11. Encaminhar o cliente ao caixa;
	12. Perguntar ao cliente se deseja informar o CPF na nota fiscal;
	13. Perguntar ao cliente a forma de pagamento desejada;
	14. Efetuar o recebimento do pagamento;
6. Fim	15. Entregar o(s) remédio(s);
	16. Entregar a nota fiscal;
	17. Encerrar o atendimento.



Fluxograma:

INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO



INTRODUÇÃO À PROGRAMAÇÃO


EXERCÍCIOS E LABORATÓRIO



--	--



Fluxograma:





2.2. Lista de exercícios 02

Para a lista de problemas abaixo, fazer o algoritmo em português coloquial, o fluxograma e o algoritmo em português estruturado (código portugal), escrito em Visualg ou Pascal, não esquecendo de descrever o objetivo, listar e usar todas as entradas e saídas:

1. Exemplo: um aluno deseja fazer um programa que, a partir de um número digitado no teclado, mostre na tela se este número é par ou ímpar.

OBJETIVO: verificar se um número é par ou ímpar.


ENTRADA(S): número a ser verificado.

SAÍDA(S): se o número é par ou ímpar.

ALGORITMO:

1. Início;
2. Obter o número a ser verificado;
3. Fazer a operação de resto de divisão entre o número a ser verificado e o número 2;
4. Se o resto da divisão for igual a 0, o número é par;
5. Caso contrário, se o resto da divisão for igual a 1, o número é ímpar;
6. Mostrar o resultado;
7. Fim.

FLUXOGRAMA:



CÓDIGO PORTUGOL:

```
algoritmo "ParImpar"
var
    Numero, Resto: inteiro
    Resultado: caracter
inicio
    escreva("Digite um numero inteiro: ")
    leia(Numero)
    Resto <- Numero MOD 2
    se Resto=0 entao
        Resultado <- "PAR"
    senao
        Resultado <- "IMPAR"
    fimse
    escreva("O numero eh: ", Resultado)
fimalgoritmo
```

2. Um analista do mercado financeiro deseja fazer um programa que calcule o rendimento de uma caderneta de poupança em um determinado mês, a partir do saldo atual e da rentabilidade naquele mês.

INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO



3. Um comerciante deseja saber qual deverá ser o preço de venda de um produto, sabendo-se que a partir do preço de custo é aplicada a margem de lucro que é de 20% e sobre o total é calculado o ICMS que é de 25%.
4. Um motorista deseja calcular o consumo médio de combustível de um automóvel, a partir da distância percorrida e do total de combustível consumido.
5. Um analista do mercado financeiro deseja fazer um programa que faça a conversão de dólar para real, a partir da cotação do dia e da quantidade de dólares que se deseja converter.
6. Uma clínica médica deseja mostrar, a partir da idade de uma pessoa dada em anos, se ela é adulta (maior ou igual a 18 anos) ou se é menor de idade (menor de 18 anos).
7. Um professor deseja calcular a média de um aluno a partir de 4 notas, que variam de 0 a 10, e mostrar se ele está aprovado ou reprovado. O aluno para ser aprovado precisa obter média maior ou igual a 7, caso contrário estará reprovado.
8. Uma loja de roupas está com a seguinte promoção: produtos com valor igual ou acima de R\$ 150,00 tem 10% de desconto. Um programador foi contratado para alterar o sistema, que deverá mostrar as seguintes informações: Valor do Produto, Valor do Desconto e Valor Final.



2.3. Lista de exercícios 03

Para a lista de problemas abaixo, fazer o algoritmo em português coloquial, o fluxograma e o algoritmo em português estruturado (código portugol), escrito em Visualg ou Pascal, não esquecendo de descrever o objetivo, listar e usar todas as entradas e saídas:

1. Exemplo: uma clínica médica deseja mostrar, a partir da idade de uma pessoa dada em anos, qual a sua faixa etária, ou seja, se ela é idosa (maior ou igual a 60 anos), adulta (maior ou igual a 18 anos e menor de 60 anos) ou se é menor de idade (menor de 18 anos).

OBJETIVO: verificar a faixa etária de uma pessoa.

ENTRADA(S): idade da pessoa.

SAÍDA(S): faixa etária.

ALGORITMO (Concatenado):

Início;

Obter a idade da pessoa;

Se a idade da pessoa for maior ou igual a 60 anos, a pessoa é idosa;


Se a idade da pessoa for maior ou igual a 18 anos e menor que 60 anos, a pessoa é adulta;

Se a idade da pessoa for menor que 18 anos, a pessoa é menor de idade;

Mostrar a faixa etária;

Fim.

FLUXOGRAMA (Concatenado):



CÓDIGO PORTUGOL (Concatenado):


```
algoritmo "Idade"
var
    Idade: inteiro
    FaixaEtaria: caracter
inicio
    escreva("Digite a idade da pessoa: ")
    leia(Idade)
    se Idade >= 60 entao
        FaixaEtaria <- "Idosa"
    fimse
    se (Idade >= 18) E (Idade < 60) entao
        FaixaEtaria <- "Adulta"
    fimse
    se Idade < 18 entao
        FaixaEtaria <- "Menor"
    fimse
    escreva("A pessoa eh ", FaixaEtaria)
fimalgoritmo
```



ALGORITMO (Aninhado ou encadeado):

25. Início;
26. Obter a idade da pessoa;
27. Se a idade da pessoa for maior ou igual a 60 anos, a pessoa é idosa;
28. Caso contrário, se a idade da pessoa for maior ou igual a 18 anos, a pessoa é adulta;
29. Caso contrário, a pessoa é menor de idade;
30. Mostrar a faixa etária;
31. Fim.

FLUXOGRAMA (Aninhado ou encadeado):



CÓDIGO PORTUGOL (Aninhado ou encadeado):

```
algoritmo "Idade"
var
    Idade: inteiro
    FaixaEtaria: caracter
inicio
    escreva("Digite a idade da pessoa: ")
    leia(Idade)
    se Idade >= 60 entao
        FaixaEtaria <- "Idosa"
    senao
        se Idade >= 18 entao
            FaixaEtaria <- "Adulta"
        senao
            FaixaEtaria <- "Menor"
        fimse
    fimse
    escreva("A pessoa eh ", FaixaEtaria)
fimalgoritmo
```

2. Um professor deseja calcular a média de um aluno a partir de 4 notas, que variam de 0 a 10, e mostrar se ele está aprovado, de recuperação ou reprovado. O aluno para ser aprovado precisa obter média maior ou igual a 7. Se o aluno obter uma média maior ou igual 5 e menor que 7, ele irá para recuperação, caso contrário estará reprovado.
3. Uma loja de roupas está com a seguinte promoção: produtos com valor igual ou acima de R\$ 150,00 tem 10% de desconto, produtos com valor igual ou acima de R\$ 300,00 tem 15% de desconto e produtos com valor igual ou acima de R\$ 600,00 tem 20% de desconto. Um programador foi contratado para alterar o sistema, que deverá mostrar as seguintes informações: Valor do Produto, Valor do Desconto e Valor Final.



2.4. Lista de exercícios 04

1. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Caso-1"
02 var
03     opcao: inteiro
04 inicio
05     escreva("Digite um numero de 1 a 3: ")
06     leia(opcao)
07     escolha opcao
08         caso 1
09             escreva("Voce escolheu o numero 1.")
10         caso 2
11             escreva("Voce escolheu o numero 2.")
12         caso 3
13             escreva("Voce escolheu o numero 3.")
14     fimescolha
15 fimalgoritmo
```

2. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Caso-2"
02 var
03     opcao: inteiro
04 inicio
05     escreva("Digite um numero de 1 a 3: ")
06     leia(opcao)
07     escolha opcao
08         caso 1
09             escreva("Voce escolheu o numero 1.")
10         caso 2
11             escreva("Voce escolheu o numero 2.")
12         caso 3
13             escreva("Voce escolheu o numero 3.")
14         outrocaso
15             escreva("Voce nao escolheu um numero de 1 a 3!")
16     fimescolha
17 fimalgoritmo
```

3. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Caso-3"
02 var
03     opcao: inteiro
04 inicio
05     escreva("Digite um numero de 0 a 9: ")
06     leia(opcao)
07     escolha opcao
08         caso 0, 2, 4, 6, 8
09             escreva("Voce digitou um numero par.")
10         caso 1, 3, 5, 7, 9
11             escreva("Voce digitou um numero impar.")
12     fimescolha
13 fimalgoritmo
```



4. Alterar o programa anterior para inserir a seguinte mensagem no caso do usuário digitar uma opção inválida:

```
"Voce nao digitou um numero de 0 a 9!"
```

5. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Caso-4"
02 var
03     opcao: caracter
04 inicio
05     escreva("Digite a sigla de um Estado Brasileiro: ")
06     leia(opcao)
07     escolha opcao
08         caso "AC", "AM", "AP", "PA", "RO", "RR", "TO"
09             escreva("Este Estado pertence a Regiao Norte.")
10         caso "AL", "BA", "CE", "MA", "PB", "PE", "PI", "RN", "SE"
11             escreva("Este Estado pertence a Regiao Nordeste.")
12         caso "DF", "GO", "MS", "MT"
13             escreva("Este Estado pertence a Regiao Centro-Oeste.")
14         caso "ES", "MG", "RJ", "SP"
15             escreva("Este Estado pertence a Regiao Sudeste.")
16         caso "PR", "RS", "SC"
17             escreva("Este Estado pertence a Regiao Sul.")
18         outrocaso
19             escreva("Voce digitou uma sigla invalida!")
20     fimescolha
21 fimalgoritmo
```

6. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Caso-5"
02 var
03     opcao: caracter
04     opSoma, opSubtracao, opMultiplicacao, opDivisao: caracter
05 inicio
06     opSoma <- "+"
07     opSubtracao <- "-"
08     opMultiplicacao <- "*"
09     opDivisao <- "/"
10     escreva("Digite um operador: ")
11     leia(opcao)
12     escolha opcao
13         caso opSoma
14             escreva("Voce escolheu Soma.")
15         caso opSubtracao
16             escreva("Voce escolheu Subtracao.")
17         caso opMultiplicacao
18             escreva("Voce escolheu Multiplicacao.")
19         caso opDivisao
20             escreva("Voce escolheu Divisao.")
21     fimescolha
22 fimalgoritmo
```



7. Alterar o programa anterior para inserir a seguinte mensagem no caso do usuário digitar uma opção inválida:

```
"Voce escolheu uma opcao invalida!"
```

8. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Calculadora"
02 var
03     operador: caracter
04     opSoma, opSubtracao, opMultiplicacao, opDivisao: caracter
05     valor1, valor2: inteiro
06     resultado: real
07 inicio
08     opSoma <- "+"
09     opSubtracao <- "-"
10     opMultiplicacao <- "*"
11     opDivisao <- "/"
12     escreva("Digite o primeiro valor: ")
13     leia(valor1)
14     escreva("Digite o segundo valor: ")
15     leia(valor2)
16     escreva("Digite um operador: ")
17     leia(operador)
18     escolha operador
19     caso opSoma
20         resultado <- valor1 + valor2
21         escreva(valor1, " +", valor2, " = ", resultado:5:2)
22     caso opSubtracao
23         resultado <- valor1 - valor2
24         escreva(valor1, " -", valor2, " = ", resultado:5:2)
25     caso opMultiplicacao
26         resultado <- valor1 * valor2
27         escreva(valor1, " *", valor2, " = ", resultado:5:2)
28     caso opDivisao
29         resultado <- valor1 / valor2
30         escreva(valor1, " /", valor2, " = ", resultado:5:2)
31     fimescolha
32 fimalgoritmo
```

9. Porque a variável `resultado`, declarada na linha 06 do programa anterior, tem que ser do tipo `real`?
10. Qual erro pode ocorrer ao ser executada a seguinte linha no programa no exercício 8? Como contornar este problema?

```
29         resultado <- valor1 / valor2
```

11. Explique o que faz a seguinte linha do programa do exercício 8:

```
21         escreva(valor1, " +", valor2, " = ", resultado:5:2)
```

12. Faça o fluxograma dos programas dos exercícios 1, 2, 3, 5, 6 e 8.

INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO





3. Estrutura de repetição, vetores e matrizes

Os laboratórios desta seção tem o objetivo de

Estruturas de repetição
Vetores e matrizes



3.1. Lista de exercícios 01

1. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Contador"
02 var
03     contador: inteiro
04 inicio
05     para contador de 1 ate 10 faca
06         escreva("A variavel contador agora vale: ", contador)
07     fimpara
08 fimalgoritmo
```

2. Trocar o comando da linha 06 do exercício anterior de `escreva` para `escreval` e analisar o resultado.

3. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Contador2"
02 var
03     contador: inteiro
04 inicio
05     para contador de 1 ate 10 passo 2 faca
06         escreval("A variavel contador agora vale: ", contador)
07     fimpara
08 fimalgoritmo
```

4. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Contador3"
02 var
03     contador: inteiro
04 inicio
05     para contador de 10 ate 1 passo -1 faca
06         escreval("A variavel contador agora vale: ", contador)
07     fimpara
08 fimalgoritmo
```

5. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Contador4"
02 var
03     contador, maximo: inteiro
04 inicio
05     escreva("Digite um numero de 1 a 10: ")
06     leia(maximo)
07     para contador de 1 ate maximo faca
08         escreva("A variavel contador agora vale: ", contador)
```

INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO



09 fimpара
10 fimalgoritmo



6. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Contador5"
02 var
03     hora, minuto: inteiro
04 inicio
05     para hora de 0 ate 23 faca
06         para minuto de 0 ate 59 faca
07             escreval(hora, ":", minuto)
08         fimpara
09     fimpara
10 fimalgoritmo
```


7. Trocar o comando da linha 07 do exercício anterior de `escreval(hora, ":", minuto)` para `escreval(hora:2, ":", minuto:2)` e analisar o resultado.

8. Quantas linhas o programa do exercício 6 irá mostrar na tela?

9. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Contador6"
02 var
03     contador: inteiro
04 inicio
05     para contador de 1 ate 10 faca
06         se contador MOD 2 = 0 entao
07             escreval("O numero", contador, " eh PAR")
08         senao
09             escreval("O numero", contador, " eh IMPAR")
10         fimse
11     fimpara
12 fimalgoritmo
```

10. Fluxograma do comando *para..faca* do exercício 1:



11. Desenhar os fluxogramas dos exercícios 3, 4, 5, 6 e 9.



3.2. Lista de exercícios 02

1. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Enquanto"
02 var
03     contador: inteiro
04 inicio
05     escreva("Digite um numero menor que dez: ")
06     leia(contador)
07     enquanto contador <= 10 faca
08         escreval("Contador: ", contador)
09     fimenquanto
10 fimalgoritmo
```

2. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Enquanto2"
02 var
03     contador: inteiro
04 inicio
05     escreva("Digite um numero menor que dez: ")
06     leia(contador)
07     enquanto contador <= 10 faca
08         contador <- contador + 1
09         escreval("Contador: ", contador)
10     fimenquanto
11 fimalgoritmo
```

3. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Enquanto3"
02 var
03     opcao: inteiro
04     somador: inteiro
05 inicio
06     somador <- 0
07     escreva("Digite um numero qualquer ou zero para sair: ")
08     leia(opcao)
09     enquanto opcao <> 0 faca
10         somador <- somador + opcao
11         escreval("Somador: ", somador)
12     fimenquanto
13 fimalgoritmo
```



4. Analisar e testar o código de programa abaixo:


```
01 algoritmo "Enquanto4"
02 var
03     opcao: inteiro
04     somador: inteiro
05 inicio
06     somador <- 0
07     escreva("Digite um numero qualquer ou zero para sair: ")
08     leia(opcao)
09     enquanto opcao <> 0 faca
10         somador <- somador + opcao
11         escreval("Somador: ", somador)
12         escreva("Para sair, digite zero (ou nao): ")
13         leia(opcao)
14     fimenquanto
15 fimalgoritmo
```

5. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Enquanto5"
02 var
03     R: real
04     X, Y: inteiro
05     opcao: caracter
06 inicio
07     opcao <- "+"
08     enquanto (opcao <> "s") ou (opcao <> "S" ) faca
09         escreva("Digite o primeiro valor: ")
10         leia(X)
11         escreva("Digite o segundo valor: ")
12         leia(Y)
13         escreva("Digite um operador (+, -, * ou /) ou s para sair: ")
14         leia(opcao)
15         escolha opcao
16             caso "+"
17                 R <- X + Y
18                 escreval(X, " +", Y, " =", R)
19             caso "-"
20                 R <- X - Y
21                 escreval(X, " -", Y, " =", R)
22             caso "*"
23                 R <- X * Y
24                 escreval(X, " *", Y, " =", R)
25             caso "/"
26                 R <- X / Y
27                 escreval(X, " /", Y, " =", R)
28         fimescolha
29     fimenquanto
30 fimalgoritmo
```




6. Fluxograma do comando *enquanto..faca* do exercício 1:



7. Desenhar os fluxogramas dos exercícios 2, 3, 4 e 5.

8. **Companhia de Tecnologia de Saneamento Ambiental 2009** - Considere o seguinte fluxograma para responder às questões a seguir. Assuma que entradas a, b e c lidas sejam, respectivamente, 12, 5 e 9.



Considerando ainda o fluxograma apresentado, assinale a alternativa que apresenta quantas vezes o teste marcado com (*) na figura é executado.

- 2
- 3
- 4
- 5
- 6

Assinale a alternativa que apresenta, para o algoritmo apresentado no fluxograma, a soma dos valores de a, b e c resultantes ao final da execução do algoritmo.

- 32. 17
- 33. 19
- 34. 21
- 35. 23
- 36. 26



3.3. Lista de exercícios 03

1. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Repita"
02 var
03     contador: inteiro
04 inicio
05     escreva("Digite um numero menor que dez: ")
06     leia(contador)
07     repita
08         escreval("Contador: ", contador)
09     ate contador > 10
10 fimalgoritmo
```

2. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Repita2"
02 var
03     contador: inteiro
04 inicio
05     escreva("Digite um numero menor que dez: ")
06     leia(contador)
07     repita
08         contador <- contador + 1
09         escreval("Contador: ", contador)
10     ate contador > 10
11 fimalgoritmo
```

3. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Repita3"
02 var
03     opcao: inteiro
04     somador: inteiro
05 inicio
06     somador <- 0
07     escreva("Digite um numero qualquer ou zero para sair: ")
08     leia(opcao)
09     repita
10         somador <- somador + opcao
11         escreval("Somador: ", somador)
12     ate opcao = 0
13 fimalgoritmo
```



4. Analisar e testar o código de programa abaixo:


```
01 algoritmo "Repita4"
02 var
03     opcao: inteiro
04     somador: inteiro
05 inicio
06     somador <- 0
07     repita
08         escreva("Digite um numero qualquer ou zero para sair: ")
09         leia(opcao)
10         somador <- somador + opcao
11         escreval("Somador: ", somador)
12     ate opcao = 0
13 fimalgoritmo
```

5. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Repita5"
02 var
03     R: real
04     X, Y: inteiro
05     opcao: caracter
06 inicio
07     repita
08         escreva("Digite o primeiro valor: ")
09         leia(X)
10         escreva("Digite o segundo valor: ")
11         leia(Y)
12         escreva("Digite um operador (+, -, * ou /) ou s para sair: ")
13         leia(opcao)
14         escolha opcao
15             caso "+"
16                 R <- X + Y
17                 escreval(X, " +", Y, " =", R)
18             caso "-"
19                 R <- X - Y
20                 escreval(X, " -", Y, " =", R)
21             caso "*"
22                 R <- X * Y
23                 escreval(X, " *", Y, " =", R)
24             caso "/"
25                 R <- X / Y
26                 escreval(X, " /", Y, " =", R)
27         fimescolha
28     ate (opcao = "s") ou (opcao = "S")
29 fimalgoritmo
```



6. Fluxograma do comando *repita..ate* do exercício 1:



7. Desenhar os fluxogramas dos exercícios 2, 3, 4 e 5.



3.4. Lista de exercícios 04

Para a lista de problemas abaixo, fazer o algoritmo em português coloquial, o fluxograma e o algoritmo em português estruturado (código portugal), escrito em Visualg ou Pascal, não esquecendo de descrever o objetivo, listar e usar todas as entradas e saídas:

1. Exemplo: Um professor deseja calcular a média de um aluno a partir de 4 notas, que variam de 0 a 10, e mostrar se ele está aprovado ou reprovado. O aluno para ser aprovado precisa obter média maior ou igual a 7, caso contrário estará reprovado.

OBJETIVO: calcular a média de um aluno e mostrar se ele está aprovado ou reprovado.


ENTRADA(S): notas.

SAÍDA(S): conceito.

ALGORITMO:

1. Início;
2. Obter notas;
3. Somar as quatro notas e dividir por quatro;
4. Se a média for maior ou igual a 7, o aluno está aprovado;
5. Caso contrário, o aluno está reprovado;
6. Mostrar o conceito;
7. Fim.

FLUXOGRAMA:



CÓDIGO PORTUGOL:

```
algoritmo "Notas-1"
var
    Nota1, Nota2, Nota3, Nota4: inteiro
    Media: real
    Conceito: caracter
inicio
    escreva("Digite a Nota 1: ")
    leia(Nota1)
    escreva("Digite a Nota 2: ")
    leia(Nota2)
    escreva("Digite a Nota 3: ")
    leia(Nota3)
    escreva("Digite a Nota 4: ")
    leia(Nota4)
    Media <- (Nota1+Nota2+Nota3+Nota4)/4
    se Media >= 7 entao
        Conceito <- "Aprovado"
    senao
        Conceito <- "Reprovado"
    fimse
    escreva(Conceito)
fimalgoritmo
```

INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO



2. Exemplo com vetores: Um professor deseja calcular a média de um aluno a partir de 4 notas, que variam de 0 a 10, e mostrar se ele está aprovado ou reprovado. O aluno para ser aprovado precisa obter média maior ou igual a 7, caso contrário estará reprovado.


<p>FLUXOGRAMA:</p> <pre> graph TD Inicio([INÍCIO]) --> Read[leia N[1], N[2], N[3] e N[4]] Read --> Calc["M <- (N[1]+N[2]+N[3]+N[4])/4"] Calc --> Dec{M >= 7} Dec -- S --> ConvA[Conceito <- "Aprovado"] Dec -- N --> ConvR[Conceito <- "Reprovado"] ConvA --> Conn(()) ConvR --> Conn Conn --> Out{{Conceito}} Out --> Fim([FIM]) </pre>	<p>CÓDIGO PORTUGOL:</p> <pre> algoritmo "Notas-2" var Nota: vetor[1..4] de inteiro Media: real Conceito: caracter inicio escreva("Digite a Nota 1: ") leia(Nota[1]) escreva("Digite a Nota 2: ") leia(Nota[2]) escreva("Digite a Nota 3: ") leia(Nota[3]) escreva("Digite a Nota 4: ") leia(Nota[4]) Media <- (Nota[1]+Nota[2]+Nota[3]+Nota[4])/4 se Media >= 7 entao Conceito <- "Aprovado" senao Conceito <- "Reprovado" fimse escreva(Conceito) fimalgoritmo </pre>
---	---

3. Exemplo com vetores e estrutura de repetição: Um professor deseja calcular a média de um aluno a partir de 4 notas, que variam de 0 a 10, e mostrar se ele está aprovado ou reprovado. O aluno para ser aprovado precisa obter média maior ou igual a 7, caso contrário estará reprovado.

<p>FLUXOGRAMA:</p>	<p>CÓDIGO PORTUGOL:</p> <pre> algoritmo "Notas-3" var Nota: vetor[1..4] de inteiro i: inteiro Media: real Conceito: caracter inicio para i de 1 ate 4 faca escreva("Digite a Nota", i, ": ") leia(Nota[i]) fimpara Media <- (Nota[1]+Nota[2]+Nota[3]+Nota[4])/4 se Media >= 7 entao Conceito <- "Aprovado" senao Conceito <- "Reprovado" fimse escreva(Conceito) </pre>
---------------------------	--

INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO





4. Fazer um programa que permita ao usuário preencher um vetor de 5 posições e que em seguida mostre este mesmo vetor na ordem inversa.
5. Fazer um programa que mostre a tabuada de um número qualquer digitado pelo usuário usando vetores.
6. Fazer um programa que permita ao usuário preencher um vetor de 5 posições e que em seguida mostre este mesmo vetor na ordem inversa. O programa deverá perguntar ao usuário se ele deseja ou não continuar.
7. Fazer um programa que permita ao usuário digitar a idade de 5 pessoas, e que ao final diga qual a soma das idades e a média.
8. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Ordena"
02 var
03     valor: vetor[0..9] de inteiro
04     i, j, aux, MAX: inteiro
05 inicio
06     MAX <- 9
07
08     //Entrada
09     para i de 0 ate MAX faça
10         escreva("Digite o valor para posicao ", i+1:2, ": ")
11         leia(valor[i])
12     fimpara
13     escreval
14
15     //Mostra valores na ordem original
16     para i de 0 ate MAX faça
17         escreval("O valor da posicao ", i+1:2, " eh", valor[i])
18     fimpara
19     escreval
20
21     //Ordena valores
22     para i de 0 ate MAX-1 faça
23         para j de i+1 ate MAX faça
24             se valor[i] > valor[j] entao
25                 aux <- valor[i]
26                 valor[i] <- valor[j]
27                 valor[j] <- aux
28             fimse
29         fimpara
30     fimpara
31
32     //Mostra valores ordenados
33     para i de 0 ate MAX faça
34         escreval("O novo valor da posicao ", i+1:2, " eh", valor[i])
35     fimpara
36     escreval
37 fimalgoritmo
```

9. Faça o fluxograma dos programas dos exercícios 4, 5, 6, 7 e 8.



3.5. Lista de exercícios 05

Para a lista de problemas abaixo, fazer o algoritmo em português coloquial, o fluxograma e o algoritmo em português estruturado (código portugal), escrito em Visualg ou Pascal, não esquecendo de descrever o objetivo, listar e usar todas as entradas e saídas:

1. Fazer um programa que permita ao usuário preencher uma matriz 3 por 3 e que em seguida a mostre na tela.
2. Fazer um programa que permita ao usuário preencher uma matriz 4 por 4 e que em seguida mostre a soma total de cada linha.
3. Fazer um programa que permita ao usuário preencher uma matriz 4 por 4 e que em seguida mostre a soma total de cada coluna.
4. Fazer um programa que permita ao usuário preencher duas matrizes, uma 3 por 2 e outra 2 por 3, e que em seguida mostre o resultado da multiplicação entre elas. Exemplo:

$$\begin{bmatrix} 3 & 9 \\ 4 & 5 \\ 7 & 2 \end{bmatrix} \times \begin{bmatrix} 4 & 6 & 3 \\ 1 & 8 & 2 \end{bmatrix} = \begin{bmatrix} (3.4+9.1) & (3.6+9.8) & (3.3+9.2) \\ (4.4+5.1) & (4.6+5.8) & (4.3+5.2) \\ (7.4+2.1) & (7.6+2.8) & (7.3+2.2) \end{bmatrix} = \begin{bmatrix} 21 & 90 & 27 \\ 21 & 64 & 22 \\ 30 & 58 & 25 \end{bmatrix}$$

5. Fazer um programa que permita ao usuário preencher duas matrizes, uma 3 por 2 e outra 2 por 3, e que em seguida mostre o resultado da multiplicação entre elas. O programa deverá perguntar ao usuário se ele deseja ou não continuar.



6. Analisar e testar o código de programa abaixo:

```
01 algoritmo "Extenso"
02 var
03     extenso: vetor[0..2, 0..10] de caracter
04     numero: inteiro
05 inicio
06     extenso[0,0] <- "Zero"
07     extenso[0,1] <- "Um"
08     extenso[0,2] <- "Dois"
09     extenso[0,3] <- "Tres"
10     extenso[0,4] <- "Quatro"
11     extenso[0,5] <- "Cinco"
12     extenso[0,6] <- "Seis"
13     extenso[0,7] <- "Sete"
14     extenso[0,8] <- "Oito"
15     extenso[0,9] <- "Nove"
16     extenso[0,10] <- "Dez"
17     extenso[1,1] <- "Onze"
18     extenso[1,2] <- "Doze"
19     extenso[1,3] <- "Treze"
20     extenso[1,4] <- "Catorze"
21     extenso[1,5] <- "Quinze"
22     extenso[1,6] <- "Dezesseis"
23     extenso[1,7] <- "Dezessete"
24     extenso[1,8] <- "Dezoito"
25     extenso[1,9] <- "Dezenove"
26     extenso[2,2] <- "Vinte"
27     extenso[2,3] <- "Trinta"
28     extenso[2,4] <- "Quarenta"
29     extenso[2,5] <- "Cinquenta"
30     extenso[2,6] <- "Sessenta"
31     extenso[2,7] <- "Setenta"
32     extenso[2,8] <- "Oitenta"
33     extenso[2,9] <- "Noventa"
34     escreva("Digite um numero de 0 a 99: ")
35     leia(numero)
36     se numero < 100 entao
37         se numero <= 10 entao
38             escreva(extenso[0,numero])
39         senao
40             se (numero > 10) E (numero < 20) entao
41                 escreva(extenso[1,numero-10])
42             senao
43                 se (numero MOD 10) <> 0 entao
44                     escreva(extenso[2,numero DIV 10], " e ", extenso[0,numero-
                                                                (numero DIV 10)*10])
45                 senao
46                     escreva(extenso[2,numero DIV 10])
47             fimse
48         fimse
49     fimse
50     escreva("Numero invalido!")
51     fimse
52 fimalgoritmo
```

7. Faça o fluxograma dos programas dos exercícios 1, 2, 3, 4, 5 e 6.

INTRODUÇÃO À PROGRAMAÇÃO

EXERCÍCIOS E LABORATÓRIO





3.6. Lista de exercícios 06

PARTE I – Cálculo do Desconto da Previdência Social

O que é Previdência Social?

A Previdência Social é um seguro que garante a renda do contribuinte e de sua família, em casos de doença, acidente, gravidez, prisão, morte e velhice. Oferece vários benefícios que juntos garantem tranquilidade quanto ao presente e em relação ao futuro assegurando um rendimento seguro. Para ter essa proteção, é necessário se inscrever e contribuir todos os meses.

Fonte: Ministério da Previdência Social

Como calcular?

Dada a Tabela de Contribuição Mensal abaixo, o cálculo do desconto dá-se da seguinte maneira: uma vez obtido o Salário Contribuição (salário base mais adicionais, etc.), deve-se compará-lo com a tabela, a fim de se conhecer a Alíquota. Uma vez conhecida a Alíquota, calcula-se o Desconto da Previdência da seguinte forma:

$$\text{Desconto da Previdência} = \text{Salário Contribuição} \times \text{Alíquota}$$

Observação: note que para o Salário Contribuição acima de R\$ 3.916,20, o valor do desconto é fixo em R\$ 430,78.

Tabela de contribuição mensal para segurados empregados para o cálculo mensal da Previdência Social, vigente a partir de 1º de janeiro de 2012

Salário Contribuição	Alíquota para fins de recolhimento ao INSS
Até R\$ 1.174,86	8%
De R\$ 1.174,87 até R\$ 1.958,10	9%
De R\$ 1.958,11 até R\$ 3.916,20	11%
Acima de R\$ 3.916,20	R\$ 430,78



Exercícios

1. Fazer um programa que calcule o desconto da previdência social.

OBJETIVO: Calcular o desconto da previdência social de acordo com a tabela de contribuição mensal.

ENTRADA(S): Salário Contribuição.


SAÍDA(S): Desconto da Previdência.

ALGORITMO:

2. Início;
3. Obter o Salário Contribuição;
4. Se o Salário Contribuição for menor ou igual a primeira faixa, o valor do Desconto da Previdência será igual ao Salário Contribuição vezes a Alíquota da primeira faixa;
5. Caso contrário, se o Salário Contribuição for menor ou igual a segunda faixa, o valor do Desconto da Previdência será igual ao Salário Contribuição vezes a Alíquota da segunda faixa;
6. Caso contrário, se o Salário Contribuição for menor ou igual a terceira faixa, o valor do Desconto da Previdência será igual ao Salário Contribuição vezes a Alíquota da terceira faixa;
7. Caso contrário, o valor do Desconto da Previdência será igual a terceira faixa vezes a Alíquota da terceira faixa;
8. Mostrar o valor do Desconto da Previdência;
9. Fim.



FLUXOGRAMA:





CÓDIGO:

```
01 algoritmo "DescontoPrevidencia"
02 var
03     SalarioContribuicao: real
04     PSFaixa1, PSFaixa2, PSFaixa3: real
05     PSAliquota1, PSAliquota2, PSAliquota3: real
06     DescontoPS: real
07 inicio
08     PSFaixa1 <- 1174.86
09     PSFaixa2 <- 1958.1
10     PSFaixa3 <- 3916.2
11     PSAliquota1 <- 0.08
12     PSAliquota2 <- 0.09
13     PSAliquota3 <- 0.11
14     escreva("Digite o Salario Contribuicao: R$ ")
15     leia(SalarioContribuicao)
16     se SalarioContribuicao <= PSFaixa1 entao
17         DescontoPS <- SalarioContribuicao*PSAliquota1
18     senao
19         se SalarioContribuicao <= PSFaixa2 entao
20             DescontoPS <- SalarioContribuicao*PSAliquota2
21         senao
22             se SalarioContribuicao <= PSFaixa3 entao
23                 DescontoPS <- SalarioContribuicao*PSAliquota3
24             senao
25                 DescontoPS <- PSFaixa3*PSAliquota3
26         fimse
27     fimse
28     fimse
29     escreva("Desconto da Previdencia: R$ ", DescontoPS:10:2)
30 fimalgoritmo
```

2. Refazer o programa do exercício 1 usando vetores e/ou matrizes.
3. Refazer o programa do exercício 1 de modo que seja perguntado ao usuário se ele deseja ou não realizar um novo cálculo.



PARTE II – Cálculo do Desconto do Imposto de Renda Retido na Fonte

O que é Imposto de Renda?

O imposto sobre a renda ou imposto sobre o rendimento é um imposto existente em vários países, em que cada pessoa ou empresa é obrigada a deduzir uma certa porcentagem de sua renda média anual para o governo federal. Esta porcentagem pode variar de acordo com a renda média anual, ou pode ser fixa em uma dada porcentagem.

Fonte: pt.wikipedia.org

Como calcular?

Dada a Tabela Progressiva abaixo, o cálculo do desconto dá-se da seguinte maneira: em primeiro lugar, deve-se obter o Salário Base do IR de acordo com a seguinte fórmula

$$\text{Salário Base IR} = \text{Salário Contribuição} - \text{Desconto da Previdência} - \text{Dedução por Dependente} \times \text{Quantidade de Dependentes}$$

Conhecendo-se então o Salário Base IR, deve-se compará-lo com a tabela, a fim de se conhecer a Alíquota e a Parcela a deduzir do imposto. Uma vez conhecidos estes valores, calcula-se o Desconto IRRF da seguinte forma:

$$\text{Desconto IRRF} = \text{Salário Base IR} \times \text{Alíquota} - \text{Parcela a Deduzir}$$

Tabela Progressiva para o cálculo mensal do IRRF vigente a partir de 1º de janeiro de 2012

Base de cálculo mensal	Alíquota	Parcela a deduzir do imposto
Até R\$ 1.637,11	-	-
De R\$ 1.637,12 até R\$ 2.453,50	7,5%	R\$ 122,78
De R\$ 2.453,51 até R\$ 3.271,38	15,0%	R\$ 306,80
De R\$ 3.271,39 até R\$ 4.087,65	22,5%	R\$ 552,15
Acima de R\$ 4.087,65	27,5%	R\$ 756,53

Dedução por dependente: R\$ 164,56



Exercícios

1. Fazer um programa que calcule o desconto do imposto de renda.

OBJETIVO: Calcular o desconto do imposto de renda de acordo com a tabela progressiva.

ENTRADA(S): Salário Contribuição, Desconto da Previdência e Quantidade de Dependentes.


SAÍDA(S): Desconto do Imposto de Renda.

ALGORITMO:

1. Início;
2. Obter o Salário Contribuição;
3. Obter o Desconto da Previdência;
4. Obter a Quantidade de Dependentes;
5. Calcular o Salário Base IR, que será igual ao Salário Contribuição menos o Desconto da Previdência menos a Dedução por Dependente vezes a Quantidade de Dependentes;
6. Se o Salário Base IR for menor ou igual a primeira faixa, o valor do Desconto do Imposto de Renda será igual zero;
7. Caso contrário, se o Salário Base IR for menor ou igual a segunda faixa, o valor do Desconto do Imposto de Renda será igual ao Salário Base IR vezes a Alíquota da segunda faixa menos a Parcela a Deduzir da segunda faixa;
8. Caso contrário, se o Salário Base IR for menor ou igual a terceira faixa, o valor do Desconto do Imposto de Renda será igual ao Salário Base IR vezes a Alíquota da terceira faixa menos a Parcela a Deduzir da terceira faixa;
9. Caso contrário, se o Salário Base IR for menor ou igual a quarta faixa, o valor do Desconto do Imposto de Renda será igual ao Salário Base IR vezes a Alíquota da quarta faixa menos a Parcela a Deduzir da quarta faixa;
10. Caso contrário, o valor do Desconto do Imposto de Renda será igual ao Salário Base IR vezes a Alíquota da quinta faixa menos a Parcela a Deduzir da quinta faixa;
11. Mostrar o valor do Desconto do Imposto de Renda;
12. Fim.



FLUXOGRAMA:





CÓDIGO:

```
01 algoritmo "DescontoImpostoRenda"  
02 var  
03     SalarioContribuicao: real  
04
```

2. Refazer o programa do exercício 1 usando vetores e/ou matrizes.
3. Refazer o programa do exercício 1 de modo que seja perguntado ao usuário se ele deseja ou não realizar um novo cálculo.



PARTE III – Folha de Pagamento

Exercícios

1. Fazer um programa que calcule o salário líquido de um funcionário após os descontos da previdência social e do imposto de renda.

OBJETIVO:


ENTRADA(S):

SAÍDA(S):

ALGORITMO:

1. Início;
- 2.

FLUXOGRAMA:



CÓDIGO:

```
01 algoritmo "FolhaPagamento"  
02 var  
03     SalarioContribuicao: real  
04
```